Vakioiden variointi kolmannen kertaluvun yhtälölle

Olkoon tarkasteltavana kolmannen kertaluvun lineaarinen epähomogeeninen differentiaaliyhtälö

```
> diffyht:= (x-1)*diff(y(x), x$3)-x*diff(y(x), x$2)+diff(y(x),
  x)=exp(x^2);
```

$$diffyht := (x-1)\left(\frac{d^3}{dx^3}\mathrm{y}(x)\right) - x\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right) + \left(\frac{d}{dx}\mathrm{y}(x)\right) = \mathbf{e}^{(x^2)}$$

Vastaava homogeeniyhtälö on

```
> homogyht:= lhs(diffyht)=0;
```

$$homogyht := (x-1)\left(\frac{d^3}{dx^3}\mathrm{y}(x)\right) - x\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right) + \left(\frac{d}{dx}\mathrm{y}(x)\right) = 0$$

Tämän ratkaisut ovat suhteellisen yksinkertaiset ja pienellä pohdiskelulla arvattavissa. Jos y:n paikalle sijoitetaan eksponenttifunktio, niin yhtälö toteutuu:

```
> subs(y(x)=exp(x), homogyht):
  simplify(%);
```

$$0 = 0$$

Samoin käy, jos sijoitetaan funktio x^2:

```
> subs(y(x)=x^2, homogyht):
  simplify(%);
```

$$0 = 0$$

Kolmanneksi perusratkaisuksi sopii vakiofunktio, joka kaikkialla saa arvon 1:

```
> subs(y(x)=1, homogyht):
  simplify(%);
```

$$0 = 0$$

Homogeeniyhtälön perusjärjestelmä on siten

```
> perusjarj:= [exp(x), x^2, 1];
```

$$perusjarj := [\mathbf{e}^x, x^2, 1]$$

`Linalg`-paketista löytyvällä `multiply`-komennolla voidaan laskea matriisituloja. Kun kyseessä on kaksi vektoria, on tuloksena pistetulo. Ladataan paketti käyttöön.

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and
```

```
unprotected
```

Homogeeniyhtälön yleinen ratkaisu on:

```
> homogrtk:= multiply(perusjarj, [_C1, _C2, _C3]);
```

$$homogrtk := \mathbf{e}^x\ \_C1 + x^2\ \_C2 + \_C3$$

Epähomogeenisen yhtälön yksittäisratkaisu voidaan hakea vakioiden varioinnilla, jolloin yrite on

```
> variot:= [u(x), v(x), w(x)];
```

$$variot := [\mathrm{u}(x), \mathrm{v}(x), \mathrm{w}(x)]$$

```
> yr0:= multiply(variot, perusjarj);
```

$$yr0 := \mathrm{u}(x)\,\mathbf{e}^x + \mathrm{v}(x)\,x^2 + \mathrm{w}(x)$$

Tämä sijoitetaan differentiaaliyhtälöön ja pyritään määrittämään sellaiset funktiot u(x), v(x) ja w(x), että yhtälö toteutuu. Laskua yksinkertaistetaan kuten toisen kertaluvun yhtälön tapauksessakin asettamalla sopivia lisäehtoja. Toisen kertaluvun tapauksessa näitä on yksi, kolmannen kertaluvun tapauksessa kaksi.

Kerätään derivaatat listaksi:

```
> derivaatat:= diff(variot, x);
```

$$derivaatat := \left[\frac{d}{dx}\,\mathrm{u}(x), \frac{d}{dx}\,\mathrm{v}(x), \frac{d}{dx}\,\mathrm{w}(x)\right]$$

Muodostetaan yritteen derivaatta

```
> der1:= diff(yr0, x);
```

$$der1 := \left(\frac{d}{dx}\,\mathrm{u}(x)\right)\mathbf{e}^x + \mathrm{u}(x)\,\mathbf{e}^x + \left(\frac{d}{dx}\,\mathrm{v}(x)\right)x^2 + 2\,\mathrm{v}(x)\,x + \left(\frac{d}{dx}\,\mathrm{w}(x)\right)$$

ja yksinkertaistetaan sitä asettamalla funktioiden u, v ja w derivaattoja sisältävien termien summa nollaksi:

```
> termit1:= [seq(coeff(der1, derivaatat[k]), k=1..3)];
```

$$termit1 := [\mathbf{e}^x, x^2, 1]$$

```
> nollatermi1:= multiply(termit1, derivaatat);
```

$$nollatermi1 := \left(\frac{d}{dx}\,\mathrm{u}(x)\right)\mathbf{e}^x + \left(\frac{d}{dx}\,\mathrm{v}(x)\right)x^2 + \left(\frac{d}{dx}\,\mathrm{w}(x)\right)$$

Tällöin derivaatta on

```
> yr1:= simplify(der1, {nollatermi1=0});
```

$$yr1 := \mathrm{u}(x)\,\mathbf{e}^x + 2\,\mathrm{v}(x)\,x$$

Edellä on käytetty komentoja tarvittavan lisäehdon muodostamiseen ja derivaatan yksinkertaistamiseen. Tämä on luontevaa kirjoitettaessa ohjelmakoodia *Maple*lle, mutta interaktiivisessa laskennassa on yksinkertaisempaa poimia tarvittavat termit hiirellä.

Vastaavalla tavalla muodostetaan yritteen toinen derivaatta, asetetaan toinen lisäehto ja yksinkertaistetaan derivaattaa:

```
> der2:= diff(yr1, x);
```

$$der2 := \left(\frac{d}{dx}\mathrm{u}(x)\right)\mathbf{e}^x + \mathrm{u}(x)\,\mathbf{e}^x + 2\left(\frac{d}{dx}\mathrm{v}(x)\right)x + 2\,\mathrm{v}(x)$$

```
> termit2:= [seq(coeff(der2, derivaatat[k]), k=1..3)];
```

$$termit2 := [\mathbf{e}^x, 2\,x, 0]$$

```
> nollatermi2:= multiply(termit2, derivaatat);
```

$$nollatermi2 := \left(\frac{d}{dx}\mathrm{u}(x)\right)\mathbf{e}^x + 2\left(\frac{d}{dx}\mathrm{v}(x)\right)x$$

```
> yr2:= simplify(der2, {nollatermi2=0});
```

$$yr2 := \mathrm{u}(x)\,\mathbf{e}^x + 2\,\mathrm{v}(x)$$

Yritteen kolmas derivaatta saadaan yksinkertaisesti derivoimalla; lisäehtoja ei enää aseteta:

```
> yr3:= diff(yr2, x);
```

$$yr3 := \left(\frac{d}{dx}\mathrm{u}(x)\right)\mathbf{e}^x + \mathrm{u}(x)\,\mathbf{e}^x + 2\left(\frac{d}{dx}\mathrm{v}(x)\right)$$

Yritteen derivaatat sijoitetaan differentiaaliyhtälöön, jolloin saadaan vain funktioiden *u*, *v* ja *w* derivaattoja koskeva ehto. Funktiot itse supistuvat pois; tämä on seurausta siitä, että yrite muodostetaan homogeeniyhtälön ratkaisujen avulla.

```
> ehto:= subs({diff(y(x), x$3)=yr3, diff(y(x), x$2)=yr2,
  diff(y(x), x)=yr1, y(x)=yr0}, diffyht):
  factor(%);
```

$$\left(\left(\frac{d}{dx}\mathrm{u}(x)\right)\mathbf{e}^x + 2\left(\frac{d}{dx}\mathrm{v}(x)\right)\right)(x-1) = \mathbf{e}^{(x^2)}$$

Derivaatat saadaan ratkaistuiksi tästä ehdosta ja aiemmin asetetuista lisäehdoista:

```
> derivrtk:= solve({ehto, nollatermi1=0, nollatermi2=0},
  {derivaatat[]}):
  simplify(%);
```

$$\left\{\frac{d}{dx}\mathrm{v}(x) = -\frac{1}{2}\,\frac{\mathbf{e}^{(x^2)}}{-2\,x+1+x^2},\ \frac{d}{dx}\mathrm{w}(x) = \frac{1}{2}\,\frac{x\,\mathbf{e}^{(x^2)}\,(-2+x)}{-2\,x+1+x^2},\ \frac{d}{dx}\mathrm{u}(x) = \frac{x\,\mathbf{e}^{(x\,(x-1))}}{-2\,x+1+x^2}\right\}$$

Funktiot u, v ja w saadaan tämän jälkeen integroimalla. Selkeintä on laskea funktiot määrättyinä integraaleina alarajan ollessa mielivaltainen. Maplessa listojen integrointi tehdään integroimalla jokainen alkio erikseen **map**-komennon avulla:

```
> integroitavat:= subs(x=t, subs(derivrtk,derivaatat)):
  simplify(%);
```

$$\left[\frac{t\,\mathrm{e}^{(t\,(-1+t))}}{-2\,t+1+t^2}, -\frac{1}{2}\,\frac{\mathrm{e}^{(t^2)}}{-2\,t+1+t^2}, \frac{1}{2}\,\frac{t\,\mathrm{e}^{(t^2)}\,(-2+t)}{-2\,t+1+t^2}\right]$$

```
> fktrtk:= map(int, integroitavat, t=a..x):
  simplify(%);
```

$$\left[\int_a^x \frac{t\,\mathrm{e}^{(t\,(-1+t))}}{-2\,t+1+t^2}\,dt, -\frac{1}{2}\int_a^x \frac{\mathrm{e}^{(t^2)}}{-2\,t+1+t^2}\,dt, \frac{1}{2}\int_a^x \frac{t\,\mathrm{e}^{(t^2)}\,(-2+t)}{-2\,t+1+t^2}\,dt\right]$$

Esiintyviä integraaleja ei onnistuta lausumaan alkeisfunktioiden tai *Maple*n tuntemien funktioiden avulla.

Vakioiden varioinnilla on kuitenkin löydetty yksittäisratkaisu:

```
> y0:= multiply(fktrtk, perusjarj):
  simplify(%);
```

$$\int_a^x \frac{t\,\mathrm{e}^{(t\,(-1+t))}}{-2\,t+1+t^2}\,dt\,\mathrm{e}^x - \frac{1}{2}\int_a^x \frac{\mathrm{e}^{(t^2)}}{-2\,t+1+t^2}\,dt\,x^2 + \frac{1}{2}\int_a^x \frac{t\,\mathrm{e}^{(t^2)}\,(-2+t)}{-2\,t+1+t^2}\,dt$$

Ratkaisu voidaan sijoittaa differentiaaliyhtälöön ja tarkistaa, toteutuuko yhtälö:

```
> subs(y(x)=y0, diffyht):
  simplify(%);
```

$$\mathrm{e}^{(x^2)} = \mathrm{e}^{(x^2)}$$

Differentiaaliyhtälön yleinen ratkaisu on siis

```
> homogrtk+y0:
  simplify(%);
```

$$\mathrm{e}^x\,\_C1 + x^2\,\_C2 + \_C3 + \int_a^x \frac{t\,\mathrm{e}^{(t\,(-1+t))}}{-2\,t+1+t^2}\,dt\,\mathrm{e}^x - \frac{1}{2}\int_a^x \frac{\mathrm{e}^{(t^2)}}{-2\,t+1+t^2}\,dt\,x^2 + \frac{1}{2}\int_a^x \frac{t\,\mathrm{e}^{(t^2)}\,(-2+t)}{-2\,t+1+t^2}\,dt$$

```
>
```

Nimittäjien nollakohdan takia tarkastelualue on joko $x < 1$ tai $x > 1$. Integraalin alaraja on valittava siitä alueesta, jota tarkastellaan.

Lukija tutkikoon, miten *Maple* ratkaisee edellä käsitellyn yhtälön suoraan **`dsolve`**-komennolla.

Linkit

epähomogeenisen yhtälön ratkaisujoukko
vakioiden variointi toisen kertaluvun yhtälön tapauksessa
korkeampien kertalukujen lineaariyhtälöt

SKK & MS 03.01.2001